

The Making of BTEX20radialDiffusion Model
using the semi-automated Modular Program Constructor

by Gary M. Raymond and James B. Bassingthwaighte

Abstract

A model of a capillary surrounded by tissue has been generated. This model consists of a concentration of a substance which in the capillary is advected by constant flow axially, undergoes diffusion axially, is well mixed radially, exchanges with the tissue through a membrane, and in the tissue undergoes axial and radial diffusion. We have assumed a radially symmetric tissue cylinder model in which the tissue is divided into 20 equally spaced rings radially and 30 segments axially as illustrated in Figure 1.




Figure 1: The capillary has radius, r_c of 4 μM . The tissue has an outer radius, R , of 20 μM . The length of the tissue cylinder, L , is 1000 μm ,

The equations governing this system are given by

$$\partial C_p / \partial t = (-F_p \cdot L / V_p) \cdot \partial C_p / \partial x + (PS / V_p) \cdot (C_{tissue} - C_p) + \partial (D_{cap} \cdot \partial C_p / \partial x) / \partial x \quad \text{and} \quad (1)$$

$$\partial C_{tissue} / \partial t = (PS / V_{tissue}) \cdot (C_p - C_{tissue}) + \partial (D_x \cdot \partial C_{tissue} / \partial x) / \partial x + 1/r \partial (r \cdot D_r \cdot \partial C_{tissue} / \partial r) / \partial r \quad . \quad (2)$$

The variables and parameters without subscripts are C (concentrations), F (Flow), L (length), V (volumes), D (diffusion coefficients), and PS (permeability surface area product governing the exchange between the capillary and tissue).

The differential equation for the tissue is converted into equations for twenty tissue layers. A radial

exchange formulation is used which is equivalent to a second-order accurate finite difference formulation for radial diffusion.

Consider the concentration in the j^{th} annular segment of length L bounded by r_j and r_{j+1} where $r_j < r_{j+1}$. The 2nd order accurate finite difference for the change in concentration, C_j , is given as

$$\partial C_j / \partial t = \frac{D_r}{(r_j + r_{j+1})/2} \cdot (C_{j+1} - C_{j-1}) / (2 \cdot \Delta r) + D_r \cdot (C_{j+1} - 2 \cdot C_j + C_{j-1}) / (\Delta r)^2 \quad (3)$$

with D_r the radial diffusion coefficient and $\Delta r = r_{j+1} - r_j$. This can also be written as

$$\partial C_j / \partial t = \frac{(D_r / \Delta r^2 - D_r / ((r_j + r_{j+1}) \cdot \Delta r)) \cdot (C_{j-1} - C_j) + (D_r / \Delta r^2 + D_r / ((r_j + r_{j+1}) \cdot \Delta r)) \cdot (C_{j+1} - C_j)}{1} \quad (4)$$

and factored as

$$\partial C_j / \partial t = \frac{(D_r / \Delta r) \cdot (1 / \Delta r - 1 / (r_j + r_{j+1})) \cdot (C_{j-1} - C_j) + (D_r / \Delta r) \cdot (1 / \Delta r + 1 / (r_j + r_{j+1})) \cdot (C_{j+1} - C_j)}{1} \quad (5)$$

This is expanded as

$$\partial C_j / \partial t = \frac{(D_r / \Delta r) \cdot (1 / (r_{j+1} - r_j) - 1 / (r_j + r_{j+1})) \cdot (C_{j-1} - C_j) + (D_r / \Delta r) \cdot (1 / (r_{j+1} - r_j) + 1 / (r_j + r_{j+1})) \cdot (C_{j+1} - C_j)}{1} \quad (6)$$

and rearranged as

$$\partial C_j / \partial t = \frac{(D_r / \Delta r) \cdot (2 \cdot \pi \cdot r_j \cdot L / (\pi \cdot (r_{j+1}^2 - r_j^2) \cdot L)) \cdot (C_{j-1} - C_j) + (D_r / \Delta r) \cdot (2 \cdot \pi \cdot r_{j+1} \cdot L / (\pi \cdot (r_{j+1}^2 - r_j^2) \cdot L)) \cdot (C_{j+1} - C_j)}{1} \quad (7)$$

Defining the permeability as

$$P_j = (D_r / \Delta r) \quad , \quad (8)$$

and the inner surface area as

$$S = 2 \cdot \pi \cdot r_j \cdot L \quad , \quad (9)$$

the permeability-surface area product is

$$PS_j = (D_r / \Delta r) \cdot 2 \cdot \pi \cdot r_j \cdot L \quad . \quad (10)$$

The volume of the j^{th} radial segment is

$$V_j = \pi \cdot (r_{j+1}^2 - r_j^2) \cdot L \quad . \quad (11)$$

We write the standard exchange process equation as

$$\partial C_j / \partial t = (PS_j / V_j) \cdot (C_{j-1} - C_j) + (PS_{j+1} / V_j) \cdot (C_{j+1} - C_j) \quad (12)$$

with the terms as defined above. Hence the the 2nd order finite difference formulation is equivalent to writing the equation as an exchange process when permeability-surface area product, PS_j , is given by Eq. 10.

The concentration variables are renumbered C0 (capillary) and C1 (first tissue ring) through C20 (last tissue ring), PS becomes PS1, V0 is the capillary volume, V1 through V20 the tissue ring volumes. Because of the similar structure of the processes involved in formulating this process, the model has been built using the Modular Program Constructor (MPC). The original program is given here:

```

/* MODEL NUMBER: 0327
   MODEL NAME: BTEX20radialDiffusion
   SHORT DESCRIPTION: A two region model of capillary and cell with cell
   subdivided into 20 subregions with axial and radial diffusion. Time-
   dependent plots of contours and center-of-mass of material are
   calculated using MATLAB.
*/
//%GET BTEX20radialDiffusionJavaCode.txt JavaRoutines()
import nsrunit; unit conversion on;
math BTEX20radialDiffusion {
real L = 1000 um;
// INDEPENDENT VARIABLES
realDomain t s; t.min=0; t.max=300; t.delta = 0.2;
real Ndivx = 31;
realDomain x um ; x.min=0; x.max=L; x.ct = Ndivx;
//
choice MOVIE("MAKE NEW MOVIE","PLAY LAST MOVIE","NO MOVIE")=3;
private real movie = MOVIE;

// DISPLAY WHICH FRAMES
choice FRAMES("ALL","LOG TIME set nframes")=2;
real nframes=50;
real iframes = FRAMES;

// PARAMETERS
real rc = 3 um;          // capillary radius
real R = 20 um;         // tissue radius
real rho = 1 g/ml;      // density of tissue
real Wt g;              // Weight
real vc ml; vc = PI*rc^2*L; // Volume of single capillary
real vR ml; vR = PI *R^2*L; // Volume of single cap-tiss unit
real Ncap dimensionless; Ncap = (1 ml)/(vR);
//Ncap the number of capillaries comprising 1 ml=1 cm^3
Wt = vR*rho;           // weight of a single cap-tiss unit
real Vc ml/g; Vc=vc/Wt; // Total capillary volume in 1 gm of tissue
real VR ml/g; VR=vR/Wt; // Total tissue volume in 1 gm of tissue
real Fp = 1 ml/(g*min); // Flow rate
real PS = 1 ml/(g*min); // Exchange rate between capillary and tissue
real Dcap = 1E-6 cm^2/s; // Axial diffusion in capillary
real Dx = 1E-6 cm^2/s; // Axial diffusion in tissue
real Dr = 1E-7 cm^2/s; // Radial diffusion in tissue
real P cm/s;          // Permeability of membrane between cap and tissue

```

```

real Cic = 0 mM;          // Initial concentraton in cap-tiss unit
private real nc = 20;    //number of subdivisions in cell
//%COM -----
//%REPLACE %CL%      =("CodeLibrary.mod")
//%REPLACE %i_0_20%  =("#0#20")
//%REPLACE (%j_1_20% =("#1#20"), %j_0_19% =("#0#19") )
//%REPLACE (%k_2_20% =("#2#20"), %k_1_19% =("#1#19") )
//%COM -----
// -----RADIOI OF SECTIONS
real dr cm;             // Thickness of concentric tissue rings
dr = (R-rc)/(nc);
real r%i_0_20% um; r%i_0_20% = rc + dr*%i_0_20%.0;

// -----VOLUMES OF tissue rings
real V0 ml/g; V0 =PI*rc^2*L/Wt;
real V%j_1_20% ml/g; V%j_1_20% =PI*(r%j_1_20%^2-r%j_0_19%^2)*L/Wt;

// -----EXCHANGE COEFFICIENTS FROM RADIAL DIFFUSION
real PS%j_1_20% ml/(g*min);
P=PS/(2*PI*rc* L/Wt );
PS1=(P      )*2*PI*rc* L/Wt;
PS%k_2_20%=(Dr/dr)*2*PI*r%k_2_20%*L/Wt;

extern real Cin(t) mM;
// -----DEPENDENT VARIABLES C0 is the capillary concentration
real Cout(t) mM;
real C%i_0_20%(t,x) mM;

// -----INITIAL CONDITIONS
when (t=t.min) C%i_0_20% = Cic;

// -----BOUNDARY CONDITONS
//%GET %CL% flowBC ("F=Fp", "C=C0", "V=V0", "D=Dcap", "Cout=Cout", "Cin=Cin")
//%GET %CL% noFlowBC ("C=C%j_1_20%")

// -----PARTIAL DIFFERENTIAL EQUATIONS
//%COLLECT("C%i_0_20%:t")
//%GET %CL% flowDiffCalc ("F=Fp", "C=C0", "V=V0", "D=Dcap", "Cout=Cout", "Cin=Cin")
//%GET %CL% exchangeCalc ("c1=C0", "c2=C1", "v1=V0", "v2=V1", "ps=PS1")
//%GET %CL% exchangeCalc ("c1=C%k_1_19%", "c2=C%k_2_20%", "v1=V%k_1_19%", "v2=V%k_2_20%", "ps=PS%k_2_20%")
//%GET %CL% diffusionCalc ("C=C%j_1_20%", "D=Dx")

// -----ASSEMBLE RESULTS in 3D array (axial,radial,time)
realDomain RR ; RR.min=0; RR.max=R; RR.ct=45;
real Call(x,RR,t) mM;
Call = if(RR<r0) C0 else
      if(RR>=r%k_1_19% and RR<r%k_2_20%) C%k_1_19% else
      C20;
//%ENDREPLACE
//%ENDREPLACE
//%ENDREPLACE
//%ENDREPLACE
//%GET AdditionCalculations.txt OtherCalc()
}
//%GET BTEX20radialDiffusionDocumentation.txt Documentation()

```

The above 99 line program. BTEX20radialDiffusion.mpc is expanded into 722 lines of code and comments by MPC and becomes the Jsim model, BTEX20radialDiffusion.mod. The 21 partial differential equations are assembled from 62 component pieces representing the three simple models for exchange, diffusion, and flow. Java procedures interfacing with MATLAB(TM) are inserted in front of the model and additional calculations and model documentation is inserted at the end of the model.

Using Jsim's interactive graphical user interface, the user can construct graphical displays illustrated in Figure 2.




Figure 2: A flow limited case is illustrated by the green contour plots showing concentration (vertical:radial, horizontal:axial) based on $t.max=300$ seconds. The time of the solutions are (3,6,15,30,75,150, and 300 seconds.) The center of mass plot is based on the radial distance. The slope of the outflow concentration becomes mono-exponential.